命令模式

LOad [(tanklist)] load | RAT | RAH | DELta | FLOWC[:ct [MINute]] [/PRessure: atm] [/Quiet]

Puts the specified load in one or more tanks.

在一个或多个舱室中指定装载。

LOad "itemlist" = load

Sets the load fraction (based on MAX weights) for listed fixed-weight items.

设定所列出的固定装载项目的装载分数（基于最大重量）。

LOad [(tanklist)] *

Sets the nominal load fraction equal to the actual volume divided by the maximum volume for one or more tanks (useful when not intact).

为一个或多个舱室设定额定装载分数，等于实际容积除以最大容积（对不完整时尤其有用）。

LOad [(tanklist)]

Shows the current loading of one or more tanks (screen only).

显示一个或多个舱室装载（仅屏幕显示）。

LOad [(tanklist)] EDIT [/VOlume:unit] [/VDEC:places] [/WDEC:places] [/WEight[:ALL]] [/METONLY] [/BULK] [/API] [/MAXSPGR:maxspgr] [/DENFMT:n] [/NOTOL] [/HMAX:hmax[,hwarn]] [/FSM | /TRUEFSM] [/NOFSA] [/ROLL] [/NONEW] [/DRAFT@:loc] [/LTRIM[:len]] [/INCH | /FId | /FIE] [/SCOLumn: Sounding | Ullage | Fsm] [/ULlage] [/CRTPT:n] [/AUTO] [/EVAL:FRA] [/GMRA] [/NOGMra] [/NOGRAPH] [/INITSOLVE] [/OMIT: [TAnk] [WEight] [GRound] [THrust] [PUll] [CG] [LS] [AUto] [/NOGROUND] [/NOLs] [/KNLS] [/LSMACro:lsmacro] [/HELP:helpfile] [/CG[:ratio]] [/SAFEFILL:maxsafe] [/SAFEULL:maxull]

Enters the Load Editor with appropriate parameters (requires the LE module).

用适当的参数进入装载编辑器（需要 LE 模块）。

LOad [(tanklist)] STatus [TAnks | FIxed] [/VOLume:unit] [/VDEC:places] [/WEight[:ALL]] [/INCH | /FId | /FIE] [/SCOLumn: Sounding | Ullage | Fsm] [/ULlage] [/EVAL:FRA] [/SORTUP | SORTDOWN: column] [/NOHEAD] [/FORMat:f [/SAFEFILL:maxsafe] [/SAFEULL:maxull]]

Provides status output similar to the Load Editor screen.

规定输出状态，类似与装载编辑器屏幕。

参数说明

tanklist

The names of one or more tanks or groups which are to have their loads changed, displayed or edited. Tank names in the tanklist may end in an asterisk to represent all tanks whose names have the same beginning. If this parameter is omitted, the current tank selection is assumed (see the TANKS command for establishing a current tank selection).

需要改变、显示或编辑的一个或多个舱室或组的名称。舱室名后面可加上*，则表示所有以该部分名称开头的舱室。如果该参数省略，则假设选取当前舱室（见 TANKS 命令，确定当前舱室选取）。

"itemlist"

One or more descriptions of fixed-weight items (where each description is enclosed in quotation marks), or "*" for all weights with maximums assigned.

一个或多个固体重量项目的描述（每个描述都在双引号之内），或者"*"表示所有重量都为其最大重量。

load

The load value takes one of the following forms:

装载值可按以下任一形式定义：

1. fraction for a loading fraction (e.g. 0.5);
1. 装载分数（例如，0.5）；
2. WEight:n for a weight (e.g. WE:123.4);
2. 重量（例如，WE:123.4）；
3. VOlume:n for a volume (cubic units only; e.g. VO:1234.5);
3. 容积（只适用立方单位；例如，VO:1234.5）；
4. HEight:n for a Reference Point height (synonyms: HT, REFpt);
4. 参考点高度（同义词：HT, REFpt）；
5. SOunding:n for a sounding (e.g. SO:4.5; synonym: SNdg);
5. 液位测深（例如，SO:4.5；同义词：SNdg）；
6. ULlage:n for an ullage (e.g. UL:5.5);
6. 液位空高（例如，UL:5.5）；
7. INtermediate:fraction for an intermediate load fraction (e.g. IN:0.25);
7. 中间装载分数（例如，IN:0.25）；
8. GRvoid:girder,dist for a standard void depth parameter (e.g. GR:3,12).
8. 标准空档深度参数的梁和距离（例如，GR:3,12）。

RAT

Loads the tank(s) so as to diminish the trim righting arm. Requires prior weight/displacement equilibrium. Does not change draft or trim; only changes the tank loading. Takes into account trimming moment if present (but only when the Axis angle is zero; see the AXIS command).

装载舱室以减少纵向回复力臂。要求在重量平衡之前。不改变吃水或纵倾；只改变舱室装载。如果引用该参数，则考虑纵倾力矩（仅当轴线角为 0 时；见 AXIS 命令）。

RAH

Similar to RAT, but applies to heel. Loads the tank(s) so as to diminish the heel righting arm.

类似 RAT，但适用于横倾。装载舱室以减少横向回复力臂。

DELTA

Attempts to load the tank(s) so as to bring about weight/displacement equilibrium at the current draft. Does not change draft or trim; only the tank loading.

在当前吃水下装载舱室，使得重量和排水量平衡。不改变吃水或纵倾；仅改变舱室装载。

FLOWC [:ct [MINUTE]]

Causes load changes in certain tanks under certain conditions that are defined via Critical Points with FLOOD status. The subject tank must have been referenced by a CRTPT definition in its /TANK parameter, and that point is considered to be the location of the opening through which the flow takes place. The magnitude of the flow (load increment) is determined by the size of the opening given with CRTPT /OPENING, the net head at the opening location, and any ct value given with the LOAD FLOWC command. The magnitude of the load increment is calculated as,

使得特定舱室装载在特定情况下改变，该特定情况由进水点的浸水状态决定。目标舱室必须在 /TANK 参数中定义了 CRTPT，这个点将被认为是进水的开口位置。进水的多少（重量增加）由 CRTPT /OPENING 中定义的开口尺寸、开口位置静水压头、以及命令 LOAD FLOWC 中给定值共同决定。增加的装载荷计算如下：

flow = √(Head) * Opening * SGratio * ct

where the Head difference is taken in feet and SGratio is specific gravity of the source divided by the receiving tank specific gravity. For example, if the Head is 2.7432 meters (9.0 feet), the Opening is 0.5 square meters, SGratio is 1.0 and ct is 2.0, then the flow volume is 3.0 x 0.5 x 1.0 x 2.0 = 3.0 cubic meters. Note ct balances units by implicitly including √2*g) factor (where g is the acceleration of gravity), as per the Bernoulli flow equation. If MINUTE is present, then ct is taken as the number of minutes per flow step by scaling it by 481.3, which is 60 seconds per minute times √2*g) in English units (or 265.7 times the √3.28 head conversion factor in metric units).LOAD FLOWC without a ct value defaults to 1 MINUTE.

其中，HEAD 不同处在于取英尺，SGratio 为进入液体比重除以原舱容液体比重。例如，如果 Head 为 2.7432 米（9.0 英尺），Opening 0.5 平方米，SGratio 为 1，ct 为 2，那么增长载荷量为 3.0 x 0.5 x 1.0 x 2.0 = 3.0 立方米。注意根据伯努利流方程，ct 默认平衡单位包括√2*g 因子（g 为重力加速度）。如果引用 MINUTE 参数，那么 ct 取每流动级别缩放 481.3 后的分钟数，也就是英制单位时 60 秒每分钟乘以√2*g（或换算为公制单位后为 265.7 乘以√3.28）。没有 ct 的 LOAD FLOWC 默认为 1 分钟。

/PRESSURE: atm

The tank surface pressure in atmospheres at the given load (default=1 atm). Applies only for BUBBLE-type tanks. This resets the nominal load at 1 atm so the tank has the same gas bubble mass as at the specified load and pressure.

舱室表面压力为规定载荷大气压（默认 1 个大气压）。只适用于气泡类舱室。这重置额定负载为 1 个大气压，从而舱室有和指定装载和压力状态下相同的气泡量。

Note that the effective liquid load present and gas pressure are adjusted automatically so internal and external pressures balance at the reference point. If the load is given as "*", then the nominal load is set so the specified internal pressure is in effect after pressures have been balanced.

注意有效液体装载和气体压力是自动调整的，所以内部和外部压力会在参考点平衡。如果装载为"*"，则额定装载被设定，那么指定内部压力实际上是在压力平衡之后的压力。

/QUIET

Suppresses screen completion messages (e.g. "Load set in all tanks" and "All tanks empty").

阻止屏幕上出现指令完成信息（例如，"Load set in all tanks" 和 "All tanks empty"）。

EDIT

Enters the Load Editor for interactive editing of fixed weights and tank loads. Load Editor may be used in either Graphics or Text mode. Graphics mode displays a profile and plan view of the vessel showing all tanks. To force Text mode, use the /NOGRAPH parameter.

进入装载编辑器进行固体重量和舱室装载的对话式编辑。装载编辑器可用在图样或文本模式。图样模式显示侧视图及所有舱室。使用参数/NOGRAPH 则强制使用文本模式。

Note: When EDIT is specified, any of the slash parameters from the LS command may be included also. These will be passed to LS when it is invoked directly from the Load Editor. See the LS command for details on these parameters.

注意：当指定 EDIT 时，任何 LS 命令的斜线参数同样可调用。当它在装载编辑器中被直接调用时，将会传递到 LS 中。详见 LS 命令。

/VOLUME: unit

Specifies which 2-letter volume unit is to be presented, chosen from the following:

指定 2 个字母为容积单位，详情如下：

GA - gallons;

GA - 加仑；

IM - imperial gallons (synonyms: IG, GI);

IM - 英制加仑（同 IG,GI）；

CF - cubic feet;

CF - 立方英尺；

CY - cubic yards (synonym: YD);

CY - 立方英码；

CM - cubic meters;

CM - 立方米；

LI - liters;

LI - 升；

BB - 42-gallon barrels.

BB - 42- 加仑桶.

/VDEC: places

Specifies the number of decimal places to use when displaying volume values.

指定舱容数值的小数位数。

/WDEC: places

Specifies the number of decimal places to use when displaying weight values.

指定舱室重量的小数位数。

/WEIGHT [:ALL]

Causes weights to be shown rather than volumes. Weight items beginning with "+" are included if and only if /WEIGHT: ALL is specified.

显示重量而不是容积。仅当设定了/WEIGHT:ALL 时，以"+"开头的重量项目包括在内。

/METONLY

Eliminates the availability of English units for volume and density when in the Load Editor. Effective in metric mode only.

当处于装载编辑器时，消除容积和密度英制单位的有效性。仅使用公制模式。

/BULK

Makes "stowage factor" (cubic feet per long ton, or cubic meters per metric ton if /METONLY) available as a means of setting contents density in Load Editor.

让“积载因素”（立方英尺/长吨，如果设定/METONLY，则为立方米/公吨）可用，作为装载编辑器中设定容积密度的一个方法。

/API

Forces API units when in metric mode, replacing KG/CM density.

公制模式时，强制 API 单位替换 KG/CM 密度。

/MAXSPGR: maxspgr

Prevents specific gravity values greater than max spgr from being entered into the Contents field.

确保比重值不大于比重表中指定的最大比重。

/DENFMT: n

Specifies the density format to be used initially in the Load Editor. The number n corresponds to the function key number for the density units.

指定装载编辑器中的初始比重格式。数字 n 相当于密度单位的功能键。

/NOTOL

Specifies that no extrapolation be allowed when using maximum VCG data.

指定当使用最大 VCG 数据时，不允许任何推算。

/HMAX: hmax [, hwarn]

Specifies maximum heel angle beyond which Maximum VCG is "unknown" (default=none) along with optional warning heel at which Maximum VCG and VCG Margin appear in yellow.

指定最大横倾角，超过该角度会导致最大 VCG 为"unknown"（默认为 none），同时可选发出警告，最大 VCG 和 VCG Margin 显示为黄色。

/FSM

Indicates that the nominal FSM values as obtained from the tank FSM functions are to be used. (This is assumed unless /TRUEFSM appears.)

把舱室自由液面矩函数得到的值作为额定自由液面矩值。（除非引用了参数/TRUEFSM）。

/TRUEFSM

Indicates that true FSM values are to be computed regardless of tanks' present FSM functions.

把真实自由液面矩值用于计算，而不管当前舱室自由液面状态。

/NOFSA

Displays Free Surface Adjustment as "n/a" and ignores it when computing Effective VCG.

显示自由液面修正为"n/a"，并且忽略其对 VCG 的影响。

/ROLL

Causes roll period to be displayed initially instead of GM.

初始显示为横摇周期，而不是 GM。

/NONEW

Prevents creating, renaming, or deleting items.

禁止创建、重命名或删除项目。

/DRAFT@: loc

Specifies that the draft is to be reported at the given location.

显示指定位置的吃水值。

/LTRIM [:len]

Specifies that trim is to be shown as a draft difference over the given length len or the LWL if len is not given.

指定纵倾显示为一定长度下的吃水差，如果不指定长度，默认为 LWL。

/INCH

Formats soundings and ullages in inches and hundredths. This parameter (along with /FID and /FIE) has no effect in metric unit modes.

指定液位测深和液位空高为英寸和百分比形式。该参数与/FID 和/FIE 在公制单位模式下不会产生任何作用。

/FID

Formats sounding and ullages in feet and decimal inches (e.g. "2'08.4").

指定液位测深和液位空高为英尺和十进位法英寸形式（例如"2'08.4"）。

/FIE

Formats soundings and ullages in feet, inches, and eighths of inches (e.g. "2'083+" is 2 feet, 8 inches, 3 eighths, plus 1 additional sixteenth inch).

指定液位测深和液位空高为英尺、英寸、和八进位法英寸形式（例如"2'083+"表示 2 英尺 $8\frac{3.5}{8}$ 英寸）。

/SCOLUMN: SOUNDING | ULLAGE | FSM

Causes soundings (default), ullages, or FSM values to be displayed initially in the rightmost tank column.

使得液位测深（默认）、液位空高或自由液面矩的数值显示在舱室信息的最右列。

/ULLAGE

Causes ullages to be displayed initially instead of soundings (same as/SCOLUMN: ULLAGE).

使得初始显示项为液位空高，而不是液位测深（和命令/SCOLUMN:ULLAGE 相同）。

/CRTPT: n

Causes the given critical point's height to be displayed in the Load Editor header.

使得指定进水点高度显示在装载编辑器顶部。

/AUTO

Causes Load Editor to automatically solve for heel after every field entry (the same as continually pressing Ctrl-A).

使得装载编辑器在每次界面输入后，自动求解横倾（等同于连续按键 Ctrl-A）。

/EVAL: FRA

Permits entering and displaying fixed-weight LCG values as offsets from frame file codes, abbreviated from descriptions listed in the frame data for the current geometry (see LS command for more details).

允许输入和显示固体重量的 LCG 值为肋位号的偏移量，肋位缩写为当前模型中定义的肋位信息（详见 LS 命令）。

/GMRA

Causes GMt displayed in Load Editor to be derived from righting arms rather than waterplane.

使得显示在装载编辑器中的 GMt 值源自于回复力臂，而不是水线面。

/NOGMRA

Disables toggling between waterplane and RA-derived GMt by pressing Ctrl-R.

禁止通过 Ctrl-R 键使 GMt 在水线面和回复力臂取法之间进行切换。

/NOGRAPH

Turns off Graphics mode in Load Editor so that only Text is displayed.

关闭装载编辑器中的图像模式，从而只显示文字。

/INITSOLVE

Initializes heel and trim to zero along with history-dependent data before any solving controlled by Load Editor. This does not affect any explicit SOLVE commands in LE macros.

在执行任何由装载编辑器控制的求解步骤之前，初始化横倾和纵倾为零，同时也初始化其他已赋值的数据。但这并不影响 LE 宏中明确标明的 SOLVE 命令。

/OMIT: [TANK] [WEIGHT] [GROUND] [THRUST] [PULL] [CG] [LS] [AUTO]

Prevents Load Editor from entering the indicated modes using Tab or Ctrl keys. Note that the /NOGROUND and /NOLS parameters have the same effect as /OMIT: GROUND and /OMIT: LS. If /AUTO is used with /OMIT:AUTO, then automatic solving mode is locked in.

阻止使用 Tab 或 Ctrl 键进入装载编辑器的指定模式。注意参数/NOGROUND 和/NOLS 与/OMIT: GROUND 和/OMIT:LS 功能相同。如果输入/OMIT:AUTO，则锁定自动求解模式。

/NOGROUND

Omits Ground Point mode from Tab rotation.

忽略 Tab 切换中的搁浅模式。

/NOLS

Prevent the operator from running LS within Load Editor. When used, "Ctr-S Strength" is replaced in Load Editor by "Ctr-T Transfer."

阻止装载编辑器进行总纵强度运算。输入该命令，会使得装载编辑器中的"Ctr-S Strength"被"Ctr-T Transfer."替代。

/KNLS

Cause Kilonewton units to be used during Longitudinal Strength display only.

使得总纵强度显示单位为千牛顿。

/LSMACRO: lsmacro

Uses a special version of Longitudinal Strength specified in the lsmacro to be executed in response to Ctrl-S (mirroring output to any open report file), instead of running LS directly (which just writes to the screen).

执行 lsmacro 中指定的总纵强度的一种特殊形式来响应 Ctrl-S 键（镜像任何进行中的报告文件的输出），而不是直接运行 LS 命令。

/HELP: helpfile

Spawns a single VIEW of helpfile by keying Ctrl-Y.

通过输入 Ctrl-Y 重新生成一个独立的帮助界面。

/CG [:ratio]

Launches an adjoining Condition Graphics window upon Load Editor startup (if the CG module is present). The CG window remains automatically synchronized with all changes the user makes in Load Editor (even if it is closed and restarted by keying Ctrl-K). If a variable named CGPARAM has been defined, its value is used as the DISPLAY STATUS parameter to configure Condition Graphics. The CG window closes automatically upon Load Editor exit. The ratio optional subparameter (default=1) sets the CG to program window height ratio.

在装载编辑器启动时，在其上方生成一个相邻的工况图窗口（要求当前为 CG 模块）。CG 窗口自动和装载编辑器中的用户操作同步（即使该窗口被关闭后又重新用快捷键 Ctrl-K 打开，其仍然保持自动同步）。如果定义了一个名为 CGPARAM 的变量，其值将被认为是用来设定工况图的 DISPLAY STATUS 参数。当装载编辑器退出时，CG 窗口自动关闭。可选子参数 ratio（默认为 1）用来设定 CG 窗口的高比。

/SAFEFILL: maxsafe

With EDIT, causes petroleum-type tank loads that exceed maxsafe to be flagged with magenta color (and an OVERFILL headlight to flash in LEw).With STATUS/FORMAT:2, reports petroleum tank capacity at maxsafe fraction of a full tank.

通过 EDIT，使得超过最大安全装载的油舱被标记为洋红色（同时装载编辑器窗口中会出现 OVERFILL 亮显标记）。用 STATUS/FORMAT:2 命令可以显示油舱在最大安全装载时的舱容和满舱舱容的百分比的小数形式。

/SAFEULL: maxull

With EDIT, causes tank loads whose ullage is less than maxull to be flagged with magenta color (and an OVERFILL headlight to appear in LEw).With STATUS/FORMAT:2, reports tank capacity at maxull ullage.

通过 EDIT，使得液位空高小于最大空高的舱室被标记为洋红色（同时装载编辑器窗口中会出现 OVERFILL 亮显标记）。用 STATUS/FORMAT:2 命令可以显示舱室在最大空高时的舱容。

STATUS [TANKS | FIXED]

Produces a status output similar to the Load Editor screen, optionally restricted to just tank loads or fixed weights.

显示一个和装载编辑器类似的状态输出，可以选择仅显示舱室装载或固体重量。

/SORTUP | SORTDOWN: column

Sorts the STATUS report up or down along the specified column, which must be one of: DEscription, NAme, COntents, LOad, PErcent, WEight, LCg, TCg, or VCg.

对指定状态列进行升序或降序排序，这些状态列必须为 DEscription, NAme, COntents, LOad, PErcent, WEight, LCg, TCg, 或 VCg。

/NOHEAD

Omits the STATUS report header.

隐藏状态台头的显示。

/FORMAT: f

Produces alternative STATUS report formats. For f=2, reports with a tank capacity column.

二选一的状态显示格式。如果 f=2，则显示舱容列。

Operation

操作

Tank Loads

舱室装载

When a load is given (either explicitly or via the RAT, RAH or DELTA parameter, or through the Load Editor), the nominal load in the tank(s) is set accordingly. If more than one tank is indicated by a group, a tank name ending in an asterisk and/or by multiple names in the list, then all tanks matching that designation will have their loads set.

当给定了装载时(可明确指定,或通过参数 RAT, RAH 或 DELTA 定义,或通过装载编辑器输入),舱室的额定负载也被相应设定。如果需要定义多个舱室，可用‘字符*’的形式，表示所有以*之前字符开头的舱室名将被同时装载。

When a tank is intact, its actual load is the same as its nominal load. But in other modes, this may not be true. For example, when flooded, the level in the tank has nothing to do with the nominal load.

如果舱室完整，那么其实际载荷就是其额定载荷。但这并不是绝对的，例如，当舱室浸水时，舱室液面和其额定载荷就毫无关联。

译者注：当舱室浸水了，GHS 默认舱内液体流出，海水流进（除非已定义为 FROZEN 型）。

Note that a tank's nominal load setting is not affected by changing its type (see the TYPE command). The LOAD command may be given regardless of the tank's present type setting and the type remains unchanged.

注意改变舱室类型（详见 TYPE 命令），不会影响其额定载荷。命令 LOAD 与舱室当前类型设定无关。

If it is desired to set the nominal load to the same value as the actual load,

LOAD (tanklist) *

will accomplish this.

如果需要设定额定载荷即为实际载荷，以下命令即可实现：

The load parameter is similar to the loadlist parameter of the TC command, except that the TC command does not include the GRVOID feature and with the LOAD command only one load value can be given. Please see the TC command documentation for details.

参数 load 和命令 TC 中的参数 loadlist 相似，不同之处在于命令 TC 不包括 GRVOID 特性，并且命令 LOAD 只能指定一个装载值。详见命令 TC。

The GRVOID parameter automatically applies a "Standard Void Depth" (following the IMO standard) for setting the surface levels in grain holds. The form of the parameter is

GRVOID: girder, dist

where girder is the girder depth and dist is the distance from hatch to bulkhead. It calculates "Vd", the average void depth, then sets the surface in the tank to a level Vd below the current reference point. Note that the GRVOID parameter is only an aid to calculating the distance from reference point to grain surface (and thereby setting the surface level). In cases where the void depth is dictated by other considerations, the HEIGHT parameter may be more convenient.

参数 GRVOID 自动为谷物装载表面高度的设定施加“标准空档深度”（遵循 IMO 标准）。参数形式如下：GRVOID: girder, dist

其中 girder 为纵桁深度，dist 为开口到舱壁距离。这用于计算"Vd"——平均空档深度，并且把舱室中的表面设定为 Vd 低于当前参考点的深度。注意参数 GRVOID 只是用于辅助计算参考点到谷物表面的距离(据此设定表面高度)。当空档深度由其它因素指明的情况下，使用参数 HEIGHT 会更加方便。

Note that setting the load by sounding, ullage or Reference Point height does not cause that value to be maintained with heel and trim changes. It does cause the load to be fixed (by means of the sounding, etc. at the time the LOAD command is issued)

注意由测深、空高或参考点高度来设定装载时，其装载并不固定，将随横倾与纵倾的变化而改变。只是在指定命令 LOAD 时，才能使得装载固定。

The RAT parameter sets the load such that the Righting Arm in Trim is brought near zero. However, the act of changing the load in the tank(s) destroys the weight/ displacement equilibrium, necessitating that the draft or weight be adjusted (by issuing a SOLVE command, for example) before the resulting trim righting arm can be evaluated. Three or four iterations of these commands are usually required before the trim righting arm vanishes. The vessel must be in weight/displacement equilibrium before a LOAD RAT command is issued.

参数 RAT 设定装载使得纵向回复力臂接近于零。然而，舱室内装载的变化会破坏重量和排水量之间的平衡，迫使在计算纵向回复力臂之前，修正吃水或重量（例如使用命令 SOLVE）。在消

除纵向回复力臂之前，通常需要重复三到四次此类命令。在输入命令 LOAD RAT 之前，必须使得船舶的重量和排水量达到平衡。

The RAH parameter works similarly to RAT, except it applies to the righting arm in heel instead of trim.

参数 RAH 和 RAT 类似，区别在于前者应用于横向回复力臂，而后者应用于纵向回复力臂。

In a similar manner, the DELTA parameter sets the load such that the difference between weight and displacement becomes zero. But contrary to the RAT operation, this achieves weight/displacement equilibrium. Therefore, LOAD RAT and LOAD DELTA can be alternated (using different tanks) in order to achieve a given trim at a given draft.

同样的原理，参数 DELTA 设定装载使得船舶重量和排水量之间差值为零。但和 RAT 相反，这使得重量和排水量达到平衡。因此，交替使用 LOAD RAT 和 LOAD DELTA 可以达到指定吃水下的指定纵倾。

A limitation involved in using RAT and DELTA loading is that the tanks must have enough capacity (and be positioned far enough from midship for RAT) so that the required weight (and moment) changes can be made.

使用 RAT 和 DELTA 装载有一个限制条件：舱室必须有足够大的舱容（对 RAT 而言还需要舱室距舯距离足够），从而能够进行所需重量（以及力矩）的变化。

The EDIT parameter is the gateway into the Load Editor, which presents a spreadsheet style interface to the user. If no (tanklist) parameter is included and there is no current tank setting, then only fixed weights will be available for editing. If no weight had been set, macro or template LE_SETLS is executed to correct the situation (if none has been defined, a default "Set Light Ship" dialog appears instead). See the Load Editor documentation for details.

参数 EDIT 为进入装载编辑器的通道，并为用户呈现了一个电子数据表形式的界面。如果没有指定舱室列表参数，并且没有设定当前参数，那么只能对固体重量进行编辑。如果没有设定固体重量，将执行宏或模板 LE_SETLS 来修正（如果没有进行任何定义，则默认显示"Set Light Ship"对话）。详见 Load Editor。

The STATUS parameter produces a table of tank loads and/or fixed weights which appear much the same as the Load Editor screens. Both tank loads and fixed weights are shown unless the TANKS or FIXED keyword is given. GROSS and NET volumes are shown for tanks with the special "oil @ t" contents description (see the CONTENTS command). See also the STATUS command for another means of displaying tank loads and weight items.

参数 STATUS 会生成一个舱室装载和固体重量的表格，其格式和装载编辑器界面非常相似。除非指定了 TANKS 或 FIXED 关键字，该表格一般会显示所有舱室装载和固体重量状态。舱室总容积和净容积会以特殊的"oil @ t"内容描述形式显示（详见命令 CONTENTS）。另一种显示舱室装载和重量信息的方法，请参看命令 STATUS。

Tank Flow

舱内液体流动

The FLOWC calculation is not intended to be physically correct without some additional considerations. The actual flow will depend on details such as the shape of the opening and its length if it is a pipe. To accurately represent a particular case, the user must adjust the CRTPT /OPENING: size parameter to compensate for these effects. The ct parameter can incorporate a constant necessary to come to a realistic flow for a particular time interval.

在没有额外一些考虑的情况下，FLOWC 计算并不是完全正确的。实际液体流动取决于开口形状等细节，如果是管路，还需要考虑其长度。为了准确的表示一个特定情况，用户必须校正参数 CRTPT /OPENING:size 来补偿这些影响。参数 ct 能够计算某些液体在特定时间间隔内的实际液体流动。

The following sequence of commands can be used in a simulation:

以下为命令使用顺序的示例：

MACRO FL

LOAD (*) FLOWC:ct

SOLVE

STATUS

/

FL (number of steps)

FL(循环计算次数)

The number of steps to be taken will depend on the details of the arrangement. Ordinarily, it will converge toward a steady state.

可浸长度计算的循环次数取决于布置的详细情况。通常将趋向一个稳定状态。

If such critical points also name an /INSIDE tank, the flow will go from the/INSIDE tank to the subject tank, with zero net change in the total weight (see the CRTPT command). The flow reverses if the pressure at the critical point is such that the flow would go out of the tank. The specific gravity settings of the contents of the tanks involved are not affected. If more than one Critical Point references the subject tank, they all contribute to the flow. If the subject tank is FLOODED or DAMAGED, the flow is zero.

如果某进水点也命名了一个/INSIDE 舱室，液体将从/INSIDE 舱流向目标舱室，总重净变化为零（详见命令 CRTPT）。如果临界点的压力会使得液体流动到舱外，那么流向会反向。舱室内液体的比重设定将不会对其产生影响。如果目标舱室有多个进水点，这些进水点都会对液体流动产生影响。如果目标舱室浸水或破损，则液体流动为零。

Fixed-Weight Loads

固体重量装载

Fixed weights which have been defined with the /MAX parameter have a nominal maximum value which is considered to be a 100% load. The LOAD command can be

used to set the weight of these items also, but only in terms of a load fraction or percentage.

使用参数/MAX 定义的固体重量拥有最大额定装载值，即为 100%装载。也可以使用命令 LOAD 来设定这些项目的重量，但仅限于装载分数或百分比的形式。

The item description(s) must be given in full, except a lone asterisk may be used to mean "all items which have a nominal maximum value".

项目描述必须给全，除非用*表示“所有项目都为额定装载”

Display Output

显示输出

When no parameters beyond (tanklist) are given, the LOAD command lists the current nominal load fractions for all the tanks in the list. When the present fraction differs from nominal, it too is shown. With BUBBLE type tanks the pressure is also shown.

当没有引用额外参数时，命令 LOAD 显示所有舱室列表中的舱室的当前额定装载比例值。当前装载比例值与额定值不同时，也将显示。BUBBLE 类型舱室还将显示其压力。

The STATUS parameter produces a table of tank loads and/or fixed weights.

参数 STATUS 会生成一个显示舱室装载及固体重量的表格。

Nondisplay Output

非显示输出

none

无

Examples

样例

Showing the loads in all tanks:

显示所有舱室中的装载：

LOAD (*)

Setting the load in tank 123-4-1 to whatever a 3.5 sounding yields:

装载舱室 123-4-1 至测深值为 3.5 米：

LOAD (123-4-1) SND: 3.5

Setting the load to 98% in all tanks whose names begin with "FO":

装载所有以"FO"开头的舱室至 98%：

LOAD (FO*) 98%

Emptying all tanks:

清空所有舱室：

```
LOAD (*) 0
```

Putting a 0.5m depth in tanks whose names begin with BAL1:

所有以 BAL1 开头的舱室内部增加 0.5 米深：

```
TANK BAL1*
REFPT BOTTOM
LOAD REF: -0.5
```

Setting the load to give an internal waterplane 1.5 below the external one:

设置装载使得内部水线面低于外部 1.5 米：

```
REFPT * * WPL
LOAD HEIGHT: 1.5
```

Setting the load such that the internal waterplane coincides with the external:

设置装载使得内部水线面与外部一致：

```
LOAD INTERMEDIATE: 1
```

A sequence to do a load for intermediate flooding:

中间浸水的装载顺序：

```
TYPE FLOODED
CONTENTS * SEA
SOLVE    `Finds the final flooded waterplane.    `求解出侵水后最终水线面。
LOAD INTERMEDIATE .25    `Loads for a 25% flooding.    `25%侵水时的装载。
TYPE INTACT    `Restores to intact condition.    `恢复到完整状况。
```

Loading tanks whose names begin with BAL1 for a 10.5 draft:

装载以 BAL1 开头的舱室至 10.5 米吃水：

```
DRAFT @ 0 = 10.5
LOAD(BAL1*) DELTA
```

Loading tank FWFP for level trim:

装载 FWFP 舱室至纵倾为零：

```
TRIM = 0
FIX TRIM
MACRO LRAT    `Defining a macro for convenience.    `定义宏，方便后面使用。
SOLVE
LOAD(FWFP) RAT
/
```

```
LRAT(3)     `Does the SOLVE & LOAD 3 times.   `重复 SOLVE 和 LOAD 三次。
VARY TRIM | SOLVE
```

Entering the Load Editor with tank volumes shown in Gallons, Longitudinal Strength calculations turned off and ullages displayed instead of soundings:

进入装载编辑器，要求：舱容单位显示为加仑，关闭总纵强度计算，并且显示液位空高而不是也为测深：

```
LOAD (*) EDIT /VOL: GA /NOLS /ULL
```

Displaying all of the cargo tanks showing GROSS and NET barrels.

显示所有货油舱的总桶量和净桶量。

```
LOAD (CARGO*) STATUS /VOL:BBL
```

Loading FWFP and tanks beginning with BAL1 for level trim at 10.5 draft:

10.5 米吃水时，装载 FWFP 舱和以 BAL1 开头的舱至纵倾为零：

```
TRIM = 0
DRAFT @ 0 = 10.5
FIX DRAFT
MAC LL     `Defining a macro for convenience.   `定义宏方便后面使用。
LOAD(BAL1*) DELTA
LOAD(FWFP) RAT
/
LL(4)     `Repeating LL 4 times.   `重复 LL 宏四次。
VARY DRAFT | SOLVE
```

Setting the load and pressure which would exist in the air bubble at that load:

设置装载与压力，使得该舱装载与压缩空气共存：

```
TANK ABC
TYPE BUBBLE
LOAD 0.5 /PRESS: 1.5
```

Setting all fixed weights which have a maximum value to zero:

设置所有拥有最大值的固体重量为零：

```
LOAD "*" = 0
```